

Предисловие

В пособии особое внимание уделяется более глубокому по сравнению с курсом информатики освоению пакета Microsoft Office, в частности Excel и Word, а также даны начальные сведения о применении Visual Basic. В основе пособия лежат задания, которые автор предлагал студентам на лабораторных занятиях по дисциплине «Информационные технологии в экономике» на протяжении 2005–2018 годов.

В разделах компьютерного практикума автор не ставил своей целью дать полное руководство по программам Microsoft Office и Visual Basic, однако приведенных в книге указаний достаточно, чтобы студенты смогли сделать первые шаги по автоматизации офисных задач и продвижению сайтов.

Сотрудники различных компаний на основе рассмотренных в пособии базовых программ на Excel и Visual Basic могут функционально развивать их или решать собственные прикладные задачи, не обращаясь к программистам.

Читатели, у которых нет времени на самостоятельное выполнение заданий предлагаемого компьютерного практикума, могут найти все приведенные в книге примеры программ на сайте издательства www.texbook.ru.

Вторая часть пособия содержит ценные указания студентам бакалавриата и магистратуры, в данном случае МГТУ им. Н.Э. Баумана, по подготовке выпускной квалификационной работы (ВКР). В частности, рассмотрены типичные ошибки оформления ВКР и показаны способы их исправления.

Во второй части пособия также рассмотрена начальная настройка редактора Word, которая позволит значительно сократить время на форматирование документов. Эти рекомендации, несомненно, будут полезны всем читателям.

Часть I.

Компьютерный практикум

Сумма прописью

Сумма прописью в графе «Всего к оплате» должна появляться автоматически. Поскольку эта операция часто встречается в большинстве финансовых и бухгалтерских документов, полезно знать, как она выполняется.

Программу «сумма прописью» удобно разместить на отдельном одноименном листе, чтобы было проще копировать ее в другие файлы Excel (см., например, разделы «Оформление счета с помощью Excel», «Платежное поручение»). Если сумма прописью нужна дважды, например для общей суммы и НДС, просто скопируйте созданную ниже программу на еще один лист, изменив всего лишь ячейку связи A1.

Пусть исходная сумма находится в ячейке A1. В сумме необходимо сначала выделить миллионы, тысячи и единицы, затем разбить соответствующие трехзначные (включая незначащие нули) числа на отдельные цифры. Для этого введите в ячейки B1 и A2–C4 следующие формулы:

B1: =ПРАВСИМВ("000000000"&ЦЕЛОЕ(A1);9)

A2: =ЗНАЧЕН(ПСТР(B1;1;1))

A3: =ЗНАЧЕН(ПСТР(B1;2;1))

A4: =ЗНАЧЕН(ПСТР(B1;3;1))

B2: =ЗНАЧЕН(ПСТР(B1;4;1))

B3: =ЗНАЧЕН(ПСТР(B1;5;1))

B4: =ЗНАЧЕН(ПСТР(B1;6;1))

C2: =ЗНАЧЕН(ПСТР(B1;7;1))

C3: =ЗНАЧЕН(ПСТР(B1;8;1))

C4: =ЗНАЧЕН(ПСТР(B1;9;1))

Первая формула добавляет к сумме в рублях девять незначащих нулей слева и затем выделяет девять символов справа. В формуле из ячейки A2 удобно преобразовать номер ячейки B1 в абсолютную адресацию ($\$B\1) по клавише $[F4]$ и затем скопировать формулу в ячейки B3–C4 и отредактировать. Все эти формулы уже из полученного текстового значения выделяют поочередно символы и преобразуют их в цифры. Здесь ПРАВСИМВ(str;n) — функция выбора n символов справа в строке str; ПСТР(str;n;m) — функция выбора подстроки длиной m, начиная с символа с номером n; ЗНАЧЕН(*) — функция преобразования текста в число.

Далее выделим из суммы число копеек, которое принято записывать двумя цифрами: не 0, а 00; не 5, а 05 и т. д. Чтобы добиться этого, введите следующие формулы:

C1: =ЦЕЛОЕ(A1*100-ЦЕЛОЕ(A1)*100+0,5),

D1: =ПРАВСИМВ("0"&ЦЕЛОЕ(C1);2).

Первая формула выделяет число копеек из общей суммы (слагаемое 0,5 необходимо для правильного округления), вторая — добавляет, если нужно, знак нуля слева.

Следующий шаг — написать словами числа миллионов, тысяч и рублей. С этой целью наберите указанную ниже таблицу (словарик). В столбцах E–H учтены грамматические окончания мужского и женского рода, а также единичного и множественного числа.

A	B	C	D	E	F	G	H
9	сто	десять	одиннадцать	один	одна	миллион	тысяча рубль
10	двести	двадцать	двенадцать	два	две	миллиона	тысячи рубля
11	триста	тридцать	тринадцать	три	три	миллиона	тысячи рубля
12	четыриста	сорок	четырнадцать	четыре	четыре	миллиона	тысячи рубля
13	пятьсот	пятьдесят	пятнадцать	пять	пять	миллионов	тысяч рублей
14	шестьсот	шестьдесят	шестнадцать	шесть	шесть	миллионов	тысяч рублей
15	семьсот	семьдесят	семнадцать	семь	семь	миллионов	тысяч рублей
16	восемьсот	восемьдесят	восемнадцать	восемь	восемь	миллионов	тысяч рублей
17	девятьсот	девяносто	девятнадцать	девять	девять	миллионов	тысяч рублей

Далее удобно дать имена столбцам словарика. Выделите курсором столбец (например, сотни A9–A17), затем переводите курсор в левую часть строки формул (под панелью инструментом, там, где вы видите номера ячеек), нажмите левую кнопку мыши, введите слово «сотни» и нажмите $[Enter]$. Затем повторите эту операцию для всех столбцов. В приведенных ниже формулах использованы следующие имена: «сотни», «десятки», «надцать» (числа от 11 до 19), «единицам» (единицы мужского рода), «единицыж» (единицы женского рода), «миллионы», «тысячи», «рубли». Удалить неправильно введенное имя можно с помощью окна Присвоить, которое вызывается командой Имя в меню Вставка.

Теперь научимся складывать названия чисел из цифр. Для этого сначала введите в ячейку A5 следующую формулу (набирать в одной строке!):

```
=ЕСЛИ(A2=0;" ";ИНДЕКС(сотни;A2;1)&"_")
&ЕСЛИ(И(A3=1;A4>0);ИНДЕКС(надцать;A4;1);
ЕСЛИ(A3=0;" ";ИНДЕКС(десятки;A3;1)&"_")
&ЕСЛИ(A4=0;" ";ИНДЕКС(единицам;A4;1)))
```

Знак & здесь обозначает *конкатенацию*, или соединение, символов, функция «И» выполняет проверку совпадения условий, функция ИНДЕКС(array;row;col) выполняет выбор элемента из массива «array» по номеру строки «row» и номеру столбца «col». Эта сложная на первый взгляд формула расшифровывается следующим образом:

если число сотен (A2) равно 0, то к строке ничего не добавлять (" "), иначе выбрать в столбце «сотни» слово, соответствующее сотням, и добавить к нему пробел ("_");

если число десятков (A3) равно 1 и число единиц (A4) не равно 0, то выбрать в столбце «надцать» слово, соответствующее числу единиц, иначе:

если число десятков (A3) равно 0, то к строке ничего не добавлять (" "), иначе выбрать в столбце «десятки» слово, соответствующее числу десятков и добавить к нему пробел ("_");

если число единиц (A4) равно 0, то к строке ничего не добавлять (" "), иначе найти в столбце «единицам» слово, соответствующее числу единиц.

Скопируйте эту формулу (автозаполнением) в ячейки B5 и C5 и поменяйте в ячейке B5 название столбца единиц на «единицыз».

Затем выберем грамматические формы слов «миллионы», «тысячи» и «рубли». Введите в ячейки A6, B6 и C6 следующие формулы (удобно формулу из ячейки A6 скопировать в B6 и C6, а затем отредактировать):

```
A6: =ЕСЛИ(ИЛИ(A3=0;A4=1);F17;ИНДЕКС(миллионы;A4;1)),
B6: =ЕСЛИ(ИЛИ(B3=0;B4=1);G17;ИНДЕКС(тысячи;B4;1)),
C6: =ЕСЛИ(ИЛИ(C3=0;C4=1);H17;ИНДЕКС(рубли;C4;1))
```

Здесь «ИЛИ» — функция, получающая значение ИСТИНА, если одно из условий верно. Смысл формул состоит в следующем: если в трехзначном числе нет единиц (A3=0) или есть числа от 11 до 19 (A4=1), то грамматическая форма соответствует множественному числу (слова «миллионов», «тысяч», «рублей» берутся из ячеек F17, G17, H17), иначе необходимо выбрать грамматическую форму слова из соответствующего столбца по числу единиц.

Теперь все готово для формирования суммы прописью. Это выполняется с помощью формулы в ячейке А7 (вновь одна строка!):

=ЕСЛИ(ДЛСТР(А5)=0;" ";А5&" "&А6&" ")

&ЕСЛИ(ДЛСТР(В5)=0;" ";В5&" "&В6&" ")

&ЕСЛИ(ДЛСТР(С5)=0;С6;С5&" "&С6)&" "

Другими словами, проверяем, есть ли разряды миллионов, тысяч и единиц (длины соответствующих строк не равны нулю), и соединяем (с помощью конкатенации) все полученные слова в одно предложение.

Принято сумму прописью начинать с прописной буквы. Заменить первую букву на прописную, а также добавить число копеек можно с помощью следующей формулы в ячейке А7:

=ПРОПИСН(ЛЕВСИМВ(А7;1))&ПРАВСИМВ(А7;ДЛСТР(А7)-1)&" "&

Д1&" коп. ".

Здесь ПРОПИСН — функция преобразования букв из строчных в прописные; ЛЕВСИМВ(str;n) — функция выбора n символов слева в строке str; ДЛСТР(str) — число символов (длина) строки str.

Таким образом, в ячейках А1–Н17 теперь находится первая программа на Excel — получения суммы прописью.

Запишем в ячейке С16 бланка простую формулу =В25. Последняя проблема: сумма может быть слишком длинной и не помещаться по ширине бланка. Для ее решения выделите диапазон ячеек С15–G16, нажмите правую кнопку мыши и в окне **Выравнивание** «взведите» два флажка: **переносить по словам** и **объединение ячеек**. Чтобы длинная сумма была видна полностью, измените ширину строки 15.

Проверьте работу программы, вводя в ячейку А1 различные числа и считывая из ячейки А7 сумму прописью.

Дополнительное задание: уберите, если необходимо, из суммы прописью копейки.